




verichains

SECURITY AUDIT OF
POLYROLL YIELD FARM
SMART CONTRACTS



Polyroll

PUBLIC REPORT

AUGUST 05, 2021

Verichains Lab

info@verichains.io

<https://www.verichains.io>

Driving Technology > Forward

ACRONYMS AND ABBREVIATIONS

NAME	DESCRIPTION
Ethereum	An open source platform based on blockchain technology to create and distribute smart contracts and decentralized applications.
Ether (ETH)	A cryptocurrency whose blockchain is generated by the Ethereum platform. Ether is used for payment of transactions and computing services in the Ethereum network
Smart contract	A computer protocol intended to digitally facilitate, verify or enforce the negotiation or performance of a contract.
Solidity	A contract-oriented, high-level language for implementing smart contracts for the Ethereum platform.
Solc	A compiler for Solidity.
EVM	Ethereum Virtual Machine
LP	Liquidity Provider

EXECUTIVE SUMMARY

This Security Audit Report was prepared by Verichains Lab on July 30, 2021, updated on August 05, 2021. We would like to thank the Polyroll team for trusting Verichains Lab in audit smart contracts. Delivering high-quality audits is always our top priority.

This audit focused on identifying security flaws in code and the design of the Polyroll core smart contracts. The scope of the audit is limited to the source code files provided to Verichains. Verichains Lab completed the assessment using manual, static, and dynamic analysis techniques.

During the audit process, the audit team had identified several vulnerable issues in the smart contracts code.

TABLE OF CONTENTS

ACRONYMS AND ABBREVIATIONS.....	2
EXECUTIVE SUMMARY.....	3
TABLE OF CONTENTS.....	4
1. MANAGEMENT SUMMARY.....	5
1.1. About Polyroll.....	5
1.2. Audit scope.....	5
1.3. Audit methodology	5
1.4. Disclaimer	6
2. AUDIT RESULT.....	7
2.1. Summary	7
2.2. Findings.....	9
2.2.1. MasterChef – Using deflationary token as LP token will lead to many vulnerabilities [MEDIUM]...	9
2.2.2. MasterChef – Missing validation for pool id in <code>set</code> function [MEDIUM]	10
2.2.3. MasterChef – Misleading comments [LOW]	11
2.2.4. MasterChef – Redundant <code>require</code> statement in the <code>updateEmissionRate</code> function [LOW].....	11
2.2.5. RollToken – Missing delegates updating when transfer tokens [CRITICAL]	12
2.2.6. RollToken – Integer overflow in <code>mint</code> function [LOW]	13
2.2.7. RollToken – Redundant <code>if</code> statement in <code>mint</code> function [LOW]	14
2.2.8. RollToken – Misleading comment of <code>delegates</code> function [LOW]	15
2.2.9. Timelock – Missing checking for valid state when cancel transaction [LOW]	15
2.2.10. Timelock – Deprecated <code>call.value(...)</code> [LOW]	16
2.2.11. RandGen – Multiple randomness requests can lead to result loss [HIGH]	17
2.3. Additional notes and recommendations.....	19
2.3.1. RandGen – Adding <code>RandomnessFulfilled</code> event.....	19
2.3.2. MasterChef – Effectively manage players in a pool.....	19
3. VERSION HISTORY.....	21
APPENDIX A: OVERVIEW FUNCTION CALL GRAPH	22

1. MANAGEMENT SUMMARY

1.1. About Polyroll

Polyroll is a decentralized casino built on Polygon Network and Chainlink. Unlike traditional casinos that operate in black boxes, Polyroll runs on smart contracts that are fair, transparent and immutable.

Polyroll is conducting a fair launch of its native token, ROLL, through yield farming. There will be no pre-sale, no private round VCs, and no pre-mine.

More information at <https://polyroll.org/>.

1.2. Audit scope

This audit focused on identifying security flaws in code and the design of the core smart contracts of Polyroll Yield Farm. It was conducted on commit 02403ae2cb0dbd083bc-6bf163d8ae00d11017977 of branch master from GitHub repository of Polyroll Yield Farm (<https://github.com/polyroll/polyroll-farm>).

Repository URL of the commit to be audited: <https://github.com/polyroll/polyroll-farm/commit/02403ae2cb0dbd083bc6bf163d8ae00d11017977>.

The updated version of report is based on the feedback of Polyroll team on July 31, 2021.

1.3. Audit methodology

Our security audit process for smart contract includes two steps:

- Smart contract codes are scanned/tested for commonly known and more specific vulnerabilities using public and RK87, our in-house smart contract security analysis tool.
- Manual audit of the codes for security issues. The contracts are manually analyzed to look for any potential problems.

Following is the list of commonly known vulnerabilities that was considered during the audit of the smart contract:

- Integer Overflow and Underflow
- Timestamp Dependence
- Race Conditions
- Transaction-Ordering Dependence
- DoS with (Unexpected) revert
- DoS with Block Gas Limit
- Gas Usage, Gas Limit and Loops
- Redundant fallback function
- Unsafe type Inference
- Reentrancy

- Explicit visibility of functions state variables (external, internal, private and public)
- Logic Flaws

For vulnerabilities, we categorize the findings into categories as listed in Table 1, depending on their severity level:

SEVERITY LEVEL	DESCRIPTION
CRITICAL	A vulnerability that can disrupt the contract functioning; creates a critical risk to the contract; required to be fixed immediately.
HIGH	A vulnerability that could affect the desired outcome of executing the contract with high impact; needs to be fixed with high priority.
MEDIUM	A vulnerability that could affect the desired outcome of executing the contract with medium impact in a specific scenario; needs to be fixed.
LOW	An issue that does not have a significant impact, can be considered as less important.

Table 1: Vulnerability severity levels

1.4. Disclaimer

Please note that security auditing cannot uncover all existing vulnerabilities, and even an audit in which no vulnerabilities are found is not a guarantee for a 100% secure smart contract. However, auditing allows discovering vulnerabilities that were unobserved, overlooked during development and areas where additional security measures are necessary.







2. AUDIT RESULT

2.1. Summary

There are three interfaces and six contracts in the Polyroll Yield Farm source codes:

- `IMasterChef` (`libs/IMasterChef.sol`): this is the interface for master chef contract which is used in the `RandGen` contract.
- `IRandGen` (`libs/IRandGen.sol`): this is the interface for random generator contract which is used in the `MasterChef` contract.
- `IReferral` (`libs/IReferral.sol`): this is the interface for referral contract which is used in the `MasterChef` contract.
- `RollToken` (`RollToken.sol`): this is contract for Polyroll Token, which is a customized ERC20 token with hard-coded maximum supply of
- `Prize` (`Prize.sol`): this is the contract which will hold and distribute the prizes for lottery games in `MasterChef` contract.
- `RandGen` (`RandGen.sol`): this is the contract that governs the requesting and receiving of random number from Chainlink VRF oracle.
- `Referral` (`Referral.sol`): this is the contract which stored the referral records that was used in `MasterChef` contract to pay referral commissions.
- `Timelock` (`Timelock.sol`): this is a timelock contract that is used to support `MasterChef` contract. This can lock the functionality of `MasterChef` for a certain amount of time.
- `MasterChef` (`MasterChef.sol`): this is the smart contract that handles functionals in Polyroll Yield Farm.

Table 2 shows the summary of findings and their statuses.

#	SUMMARY	SEVERITY	STATUS	UPDATED AT
1	MasterChef – Using deflationary token as LP token will lead to many vulnerabilities	MEDIUM	 Acknowledged	August 05, 2021
2	MasterChef – Missing validation for pool id in <code>set</code> function	MEDIUM	 Acknowledged	August 05, 2021
3	MasterChef – Misleading comments	LOW	 Acknowledged	August 05, 2021
4	MasterChef – Redundant require statement in the <code>updateEmissionRate</code> function	LOW	 Acknowledged	August 05, 2021
5	RollToken – Missing delegates updating when transfer tokens	CRITICAL	 Acknowledged	August 05, 2021
6	RollToken – Integer overflow in <code>mint</code> function	LOW	 Acknowledged	August 05, 2021

Report for Polyroll
Security Audit – Polyroll Smart Contracts

Version: 1.1 – Public Report
Date: August 05, 2021








7	RollToken – Redundant if statement in <i>mint</i> function	LOW	 Acknowledged	August 05, 2021
8	RollToken – Misleading comment of <i>delegates</i> function	LOW	 Acknowledged	August 05, 2021
9	Timelock – Missing checking for valid state when cancel transaction	LOW	 Acknowledged	August 05, 2021
10	Timelock – Deprecated <i>call.value(...)</i>	LOW	 Acknowledged	August 05, 2021
11	RandGen – Multiple randomness requests can lead to result loss	HIGH	 Acknowledged	August 05, 2021

Table 2: Findings summary

2.2. Findings

This section contains a detailed analysis of all the vulnerabilities that were discovered by the audit team during the audit process.

2.2.1. MasterChef – Using deflationary token as LP token will lead to many vulnerabilities **[MEDIUM]**

Consider the following statements in the *deposit* function of MasterChef contract:

```
225 pool.lpToken.safeTransferFrom(address(msg.sender), address(this), _amount);  
...  
234 // Update user's deposit after subtracting fee  
235 user.amount = user.amount.add(_amount).sub(depositFee);
```

The statement at line 225 calls the *safeTransferFrom* in *LpToken* contract to transfer *_amount* LP tokens from *msg.sender* to the MasterChef address. Usually, the token transfer has no token fee, so the MasterChef address will receive all *_amount* tokens from *msg.sender*. However, if the LP token is a deflationary token (token with transfer fee), then the MasterChef address will not receive all *_amount* tokens, which would cause many problems for the MasterChef contract. For example, it will make the *user.amount* (as in line 235) to be greater than the actual deposited amount (the amount MasterChef received), so deposited users will get more reward tokens than they should. The *user.amount* variable also affects the referral commission, so the users will get more referral commission.

```
218 uint256 pending =  
    user.amount.mul(pool.accRollPerShare).div(1e18).sub(user.rewardDebt);  
219 if (pending > 0) {  
220     safeRollTransfer(msg.sender, pending);  
221     payReferralCommission(msg.sender, pending);  
222 }  
  
...  
  
353 uint256 commissionAmount = _pending.mul(referralCommissionRate).div(10000);  
354  
355 if (referrer != address(0) && commissionAmount > 0) {  
356     roll.mint(_user, commissionAmount);  
357     roll.mint(referrer, commissionAmount);  
358     emit ReferralCommissionPaid(_user, referrer, commissionAmount);  
359 }
```

Therefore, if a user keeps depositing and then withdraw again and again, his Polyroll token balance will keep increasing, and the balance of MasterChef in LP token will be drained to approximately zero, which makes other users could not withdraw.

RECOMMENDATION

The current implementation of MasterChef contract does not support deflationary token, so before adding a new LP token into MasterChef for farming, this LP token must be checked to ensure that it is not a deflationary token.

In addition, to support adding deflationary token, the MasterChef contract must add safety checks to ensure that the total recorded amount of all users in a pool (sum of all *user.amount*) will be equal to the balance of MasterChef in that pool's LP token. For example, the contract can use the following code to get the actual deposited amount:

```
uint256 beforeAmount = pool.lpToken.balanceOf(address(this));  
pool.lpToken.safeTransferFrom(address(msg.sender), address(this), _amount);  
uint256 actualAmount = pool.lpToken.balanceOf(address(this)).sub(beforeAmount);
```

Then use the *actualAmount* value instead of *_amount* for calculating.

UPDATES

Polyroll team confirmed that Polyroll does not have deflationary tokens and they have no plans to support it in the future. However, as an auditor, the audit team still leave this finding here as a warning and set this finding's severity to **MEDIUM**.

2.2.2. MasterChef – Missing validation for pool id in set function **[MEDIUM]**

Consider the following code in *set* function:

```
153 function set(uint256 _pid, uint256 _allocPoint, uint16 _depositFeeBP) external  
    onlyOwner {  
154     require(_depositFeeBP <= 10000, "set: invalid deposit fee basis points");  
155     totalAllocPoint =  
        totalAllocPoint.sub(poolInfo[_pid].allocPoint).add(_allocPoint);  
156     poolInfo[_pid].allocPoint = _allocPoint;  
157     poolInfo[_pid].depositFeeBP = _depositFeeBP;  
158 }
```

In the above code, the input pool id *_pid* is not validated, so when the pool id is not exists (i.e. *_pid >= poolInfoLength*), the contract still updates the value of *totalAllocPoint* and *poolInfo*, that leads to wrong results in future reward calculations.

RECOMMENDATION

The pool id should be validated as in the following code:

```
153 function set(uint256 _pid, uint256 _allocPoint, uint16 _depositFeeBP) external  
    onlyOwner {  
154     require(_pid < poolInfoLength, "set: invalid pool id");  
155     require(_depositFeeBP <= 10000, "set: invalid deposit fee basis points");  
        totalAllocPoint =  
156 totalAllocPoint.sub(poolInfo[_pid].allocPoint).add(_allocPoint);  
157     poolInfo[_pid].allocPoint = _allocPoint;  
158     poolInfo[_pid].depositFeeBP = _depositFeeBP;  
159 }
```

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.3. MasterChef – Misleading comments [LOW]

The *randGen* variable and *getPrizeRound* function are having misleading comments.

```
95 //Roll referral contract address.  
96 IRandGen public randGen;  
  
380 // View function to see number of players in pool  
381 function getPrizeRound(uint256 _pid) external view returns (uint32) {  
382     PoolInfo storage pool = poolInfo[_pid];  
383     return pool.prizeRound;  
384 }
```

RECOMMENDATION

Update the comments to match the variable and function usages.

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.4. MasterChef – Redundant require statement in the updateEmissionRate function [LOW]

Consider the following code in the *updateEmissionRate* function:

```
319 require(rollPerBlock > FINAL_EMISSION_RATE, "updateEmissionRate: Emission rate has
    reached FINAL_EMISSION_RATE already");
...
323 require(currentIndex > lastReductionPeriodIndex, "updateEmissionRate: Wait at least
    1 day after previous update.");
324
325 // Compute new emission rate
326 uint256 newEmissionRate = rollPerBlock;
327 for (uint256 index = lastReductionPeriodIndex; index < currentIndex; ++index) {
328     newEmissionRate = newEmissionRate.mul(1e4 -
        EMISSION_REDUCTION_RATE_PER_PERIOD).div(1e4);
329 }
330 newEmissionRate = newEmissionRate < FINAL_EMISSION_RATE ? FINAL_EMISSION_RATE :
    newEmissionRate;
331 require(newEmissionRate < rollPerBlock, "updateEmissionRate: New emission rate must
    be less than current emission rate.");
```

The *require* statement at line 331 is not necessary because *newEmissionRate* has an initial value equal to *rollPerBlock*, and all the calculations at line 328 (atleast one will be executed) just reduce the *newEmissionRate* value.

RECOMMENDATION

Remove that *require* statement to make the code cleaner and save gas.

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.5. RollToken – Missing delegates updating when transfer tokens **[CRITICAL]**

RollToken allows token holders to give voting power to a delegatee. However, with the current implementation of RollToken contract, when the token holder transferred his token from his wallet, voting power of the delegate still remains the same instead of being decreased. That could lead to inflation of voting powers.

Consider the following scenario:

- Wallet A has 10 tokens in balance and don't have any voting power yet.
- A transfers 10 tokens to an empty wallet B. Now B has 10 tokens.
- B delegates its voting power to A by calling the *deLegate* function. Now A has voting power equivalent to 10 tokens.
- B transfers 10 tokens to C. Now C has 10 tokens; B don't have any tokens but A still has voting power of 10 tokens (because the voting powers were not affected while transferring tokens in the current contract implementation).
- C delegates his voting power to A. Now A has voting power equivalent to 20 tokens (10 from B and 10 from C) while the total balances of A, B and C is only 10 tokens.

By repeating the above process, a user can keep increasing his voting power with an only small initial token balance.

RECOMMENDATION

The contract must update the voting power records when transferring tokens, i.e., when *transfer* and *transferFrom* function are called. One suggestion is to call the *_moveDelegates* function in the *_beforeTokenTransfer* hook of current ERC20 contract implementation as below:

```
25 function _beforeTokenTransfer(address from, address to, uint256 amount) internal
    override {
26     _moveDelegates(_delegates[from], _delegates[to], amount);
27 }
```

Note that if the *_moveDelegates* function is called in the *_beforeTokenTransfer* hook, then it must not be called in the *mint* function in RollToken contract, since the *_beforeTokenTransfer* hook is already called in the internal *_mint* function.

UPDATES

This finding has been acknowledged by Polyroll team, as in their feedback:

“Acknowledged. We are not using voting function on the ERC20 token and do not plan on doing so in future.”

2.2.6. RollToken – Integer overflow in mint function [LOW]

Before Solidity version 0.8.0, by default, all arithmetic operations are not checked for integer overflow and underflow, so the contract must handle that itself.

Consider the following code in *mint* function:

```
17 if (_totalSupply + _amount <= MAX_SUPPLY) {
18     _mint(_to, _amount);
19     _moveDelegates(address(0), _delegates[_to], _amount);
20 } else if (_totalSupply + _amount > MAX_SUPPLY && _totalSupply <= MAX_SUPPLY) {
21     _mint(_to, MAX_SUPPLY - _totalSupply);
22     _moveDelegates(address(0), _delegates[_to], MAX_SUPPLY - _totalSupply);
23 }
```

An overflow can occur in the *_totalSupply + _amount* addition and leads to wrong result for the condition in the checking statement. For example, if *_totalSupply=1000*, *_amount=2²⁵⁶ – 1* (the maximum value for *uint256* in Solidity), then the *_totalSupply + _amount* addition will return 999, which is obviously not the result we want.

RECOMMENDATION

A simple solution is to check for integer overflow when calculate the new total supply, as in the following code:

```
13 uint _totalSupply = totalSupply();
14 require(_totalSupply < MAX_SUPPLY, "Total supply must not exceed max supply.");
15
16 uint256 _newSupply = _totalSupply + _amount;
17 require(_newSupply > _totalSupply, "Integer addition overflow.");
18
19 // Impose maximum total Supply on minting of token.
20 if (_newSupply <= MAX_SUPPLY) {
21     _mint(_to, _amount);
22     _moveDelegates(address(0), _delegates[_to], _amount);
23 } else if (_newSupply > MAX_SUPPLY && _totalSupply <= MAX_SUPPLY) {
24     _mint(_to, MAX_SUPPLY - _totalSupply);
25     _moveDelegates(address(0), _delegates[_to], MAX_SUPPLY - _totalSupply);
26 }
```

Another solution is to use the OpenZeppelin's SafeMath library¹ for arithmetic operations. If any integer overflow/underflow occurs, the transaction will be reverted. Note that from Solidity version 0.8.0, arithmetic operations will be reverted on underflow and overflow by default² without using *SafeMath* library.

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.7. RollToken – Redundant if statement in mint function [LOW]

Consider the following code of the *mint* function:

```
12 function mint(address _to, uint256 _amount) public onlyOwner {
13     uint _totalSupply = totalSupply();
14     require(_totalSupply < MAX_SUPPLY, "Total supply must not exceed max supply.");
15
16     // Impose maximum total Supply on minting of token.
17     if (_totalSupply + _amount <= MAX_SUPPLY) {
18         _mint(_to, _amount);
19         _moveDelegates(address(0), _delegates[_to], _amount);
20     } else if (_totalSupply + _amount > MAX_SUPPLY && _totalSupply <= MAX_SUPPLY) {
21         _mint(_to, MAX_SUPPLY - _totalSupply);
22         _moveDelegates(address(0), _delegates[_to], MAX_SUPPLY - _totalSupply);
23     }
24
25 }
```

¹ <https://docs.openzeppelin.com/contracts/3.x/api/math>

² <https://docs.soliditylang.org/en/v0.8.0/080-breaking-changes.html>

At line 20, the *if* statement has two conditions:

- `_totalSupply + _amount > MAX_SUPPLY`: this is always *true*, because in this case the condition at line 17 is already *false*.
- `_totalSupply <= MAX_SUPPLY`: this is always *true*, because in this case the condition at line 14 is already satisfied.

So, these two conditions are already satisfied, hence the checking is not necessary and can be removed to save gas and make the code cleaner.

RECOMMENDATION

Remove the redundant *if* statement as below:

```
17  if (_totalSupply + _amount <= MAX_SUPPLY) {
18      _mint(_to, _amount);
19      _moveDelegates(address(0), _delegates[_to], _amount);
20  } else {
21      _mint(_to, MAX_SUPPLY - _totalSupply);
22      _moveDelegates(address(0), _delegates[_to], MAX_SUPPLY - _totalSupply);
23  }
```

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.8. RollToken – Misleading comment of delegates function [LOW]

The function *delegates* in the RollToken contract is used for getting the delegatee address of *delegator*, not for delegating votes.

```
63  /**
64   * @dev Delegate votes from `msg.sender` to `delegatee`
65   * @param delegator The address to get delegatee for
66   */
67  function delegates(address delegator)
    ...
```

RECOMMENDATION

Update the comment to match the function's usage.

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.9. Timelock – Missing checking for valid state when cancel transaction [LOW]

A transaction should not be canceled if it is not in queue. Look at the current implementation of *cancelTransaction* function:

```
94 function cancelTransaction(address target, uint value, string memory signature,  
    bytes memory data, uint eta) public {  
95     require(msg.sender == admin, "Timelock::cancelTransaction: Call must come from  
        admin.");  
96  
97     bytes32 txHash = keccak256(abi.encode(target, value, signature, data, eta));  
98     queuedTransactions[txHash] = false;  
99  
100    emit CancelTransaction(txHash, target, value, signature, data, eta);  
101 }
```

If the *cancelTransaction* was called by admin, the *CancelTransaction* event will be emitted for every input transaction, even if that transaction has been executed or has not been queued yet. This behavior may lead to confusion for contract's event listeners, and wasting gas for unnecessary event emitting.

RECOMMENDATION

Only cancel transactions that are currently in queue.

```
94 function cancelTransaction(address target, uint value, string memory signature,  
    bytes memory data, uint eta) public {  
95     require(msg.sender == admin, "Timelock::cancelTransaction: Call must come from  
        admin.");  
96  
97     bytes32 txHash = keccak256(abi.encode(target, value, signature, data, eta));  
98     require(queuedTransactions[txHash], "Timelock::cancelTransaction: Transaction  
        must be queued.");  
99     queuedTransactions[txHash] = false;  
100  
101    emit CancelTransaction(txHash, target, value, signature, data, eta);  
102 }
```

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.10. Timelock – *Deprecated* `call.value(...)` [LOW]

The following code is from the `executeTransaction` function in Timelock contract:

```
122 (bool success, bytes memory returnData) = target.call.value(value)(callData);
```

From Solidity version 0.6.4, the syntax `call.value(...)` was deprecated in favor of `call{value: ...}`³. The contract should use the new syntax to avoid potential issues in future.

³ <https://github.com/ethereum/solidity/releases/tag/v0.6.4>

RECOMMENDATION

Replace the deprecated code with the new one as below:

```
122 (bool success, bytes memory returnData) = target.call{value: value}(callData);
```

UPDATES

This finding has been acknowledged by Polyroll team.

2.2.11. RandGen – Multiple randomness requests can lead to result loss [HIGH]

Current implementation of the *RandGen* contract can only store one randomness request result at a moment. Therefore, if we call the *getRandomNumber* method multiple times, whenever receiving a *fulfillRandomness* callback, the contract will override the previous result with the new result. Moreover, if the contract has multiple randomness requests in flight simultaneously, the order in which the fulfillments arrive cannot be ensured.

RECOMMENDATION

The contract should use mapping to store the random number results for each randomness request. The *settleLottery* function needs also to be modified to accept request id as an input parameter.

```
... mapping (bytes32 => bool) private _isFulfilled;
   mapping (bytes32 => uint) private _results;

48 // Callback function used by VRF Coordinator
49 function fulfillRandomness(bytes32 _requestId, uint _randomness) internal override
   {
50     _isFulfilled[_requestId] = true;
51     _results[_requestId] = randomness;
52     emit RandomnessFulfilled(_requestId, _randomness);
53 }
54
55 // SettleLottery function is separated from fulfillRandomness because
56 // Chainlink VRF has a very low gas limit and we need to call settleLottery and pay
57 gas fees ourselves.
58 function settleLottery(bytes32 _requestId) external onlyOwner {
59     require(_isFulfilled[_requestId], "Request has not been fulfilled yet.");
60     masterChef.settleLottery(_requestId, _results[_requestId]);
61 }
```

Note that if the values range of the needed random number is not as large as a *uint256* number, then we could optimize the above code to use only one mapping, thus optimize the gas usage.

UPDATES

This finding has been acknowledged by Polyroll team, as in their feedback:

“Acknowledged. We will be closing and shutting down our lottery farms, so it does not matter any more.”

2.3. Additional notes and recommendations

2.3.1. RandGen – Adding RandomnessFulfilled event

The `fulfillRandomness` function should emit an event to notify the listeners (e.g. the owner) that the current random generating request has been fulfilled and the owner can call `settleLottery` next.

Note that the Chainlink VRF Coordinator has very low gas limit (200k gas)⁴, so the `fulfillRandomness` callback function must use less gas than this limit, otherwise the transaction will fail.

RECOMMENDATION

Emit new `RandomnessFulfilled` event in the `fulfillRandomness` function.

```
... event RandomnessFulfilled(bytes32 _requestId, uint _randomness);

50 //Callback function used by VRF Coordinator
51 function fulfillRandomness(bytes32 _requestId, uint _randomness) internal override
52 {
53     requestId = _requestId;
54     randomness = _randomness;
55     emit RandomnessFulfilled(requestId, randomness);
56 }
```

2.3.2. MasterChef – Effectively manage players in a pool

In the current implementation of MasterChef contract, the struct `PoolInfo` uses two arrays (`pool.players`, `pool.playerStakes`) and one mapping (`pool.playerIds`) to store and manage players.

```
44 struct PoolInfo {
...
53     address[] players;           // List of addresses of lottery players.
54     uint256[] playerStakes;      // List of player's stakes a.k.a. player's
    contribution to lottery pool.
55     mapping(address => uint256) playerIds; // Mapping of address to indexes. Used to
    check if player has participated before this round.
56 }
```

However, that is not the most effective way to do it. The `playerIds` array is only used for checking the existence of a player. Instead, the contract could combine the `playerStakes` array and `playerIds` mapping into one (`address => uint256`) `playerStakes` mapping: if the `playerStakes[A]` is zero, then the player A has not been added to the pool yet. Note that the contract must reset values from `playerStakes` for each new round.

⁴ <https://docs.chain.link/docs/get-a-random-number/#random-number-consumer>

Additionally, in the current implementation of *settleLottery* function, after each round, the values of *pool.players* and *pool.playerStakes* will be reset by assign them to new empty array:

```
450 pool.players = new address[](0);  
451 pool.playerStakes = new uint256[](0);
```

This is not the best way to reset values for each new round, because these empty elements will eventually be filled again, so it is not an efficient gas usage. The best way is simply not to clear them. Instead, the contract should keep a separate *pool.playerCount* and manages the players like the way it manages pools using *pool.infoLength* and *pool.info* mapping. In this way, the old round players still remain in the pool (before being replaced by new players), but are ignored.

3. VERSION HISTORY

VERSION	DATE	STATUS/CHANGES	CREATED BY
0.1	July 30, 2021	Initial report	Verichains Lab
1.0	August 05, 2021	Private report	Verichains Lab
1.1	August 05, 2021	Public report	Verichains Lab

APPENDIX A: OVERVIEW FUNCTION CALL GRAPH




Figure 1: function call graph of MasterChef contract




Figure 2: function call graph of RandGen contract




Figure 3: function call graph of RollToken contract